
News  

Meetings and events 
ANS Sensorimotor Control Satellite Meeting 
Prince Of Wales Medical Research Institute 
Sydney, Australia 
January 30th – 31st 2010 

Australian Neuroscience Society/Australian Physiological Society Joint Meeting 
Darling Harbour Convention Centre 
Sydney, Australia 
January 31st – February 3rd 2010 

Latest versions of Spike2 and Signal 
Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.   

 Spike2 downloads Signal downloads 
Spike2 version 7.02 Signal version 4.06 

Spike2 version 6.13 Signal version 3.12 

Spike2 demo Signal demo 

Spike2   

Q. How can I trigger stimulators from Spike2 during a recording? 

A. Many stimulators can be triggered by a 5 volt TTL input and will generate a 
constant current or constant voltage output in response. The parameters of the 
stimulator output are usually defined using hardware controls on the stimulator. 

The Spike2 output sequencer, via the graphical editor or a text sequence file, can 
play precisely timed TTL pulses from the 1401 digital outputs during sampling 
which can be used to trigger these stimulators. Some units that do not support 5 
volt TTL compatability, such as the Grass S44 which requires a 6 volt input, can 
be triggered using a combination of a 1401 digital output and a circuit that 
produces a 9V trigger pulse (contact us for details). 

 Other constant current stimulators, such as the DS4 and DS5 units available 
from Digitimer, are controlled by an analog voltage waveform input which they 
then convert into a constant current stimulus output of proportional value. The 
1401 DAC outputs can be used to generate waveform outputs to drive these 
types of voltage-following stimulus isolators. 

CED
You can copy pulses between sections in the graphical editor to quickly set-up a series of similar stimulus protocols

david



 

  

Q. In the previous enewsletter (eNews 47) you mentioned that Signal can interface 
with the Axon MultiClamp 700A which is controlled by its own software program. 
Could you give me further details? 

A. MultiClamp telegraph support can be chosen as an alternative to the standard 
telegraph support during installation and allows Signal to read information such 
as changes to the amplifier gain, low-pass filter settings, membrane capacitance 
and series resistance direct from the MultiClamp command software. In addition, 
the clamping support features in Signal are integrated to automatically select 
between voltage clamp and current clamp, depending on the amplifier controls, 
and set up the scaling and units of the DAC used for external amplifier control. 

 
MultiClamp 700 configuration 

 The MultiClamp configuration dialog is accessed from the Port Setup tab of the 
sampling configuration. Clamp settings can be read from the amplifier software 
using the associated button in the Clamp tab of the sampling configuration. 

For more details on the telegraph configuration, see the MultiClamp section of 
the on-line help. 

Scripts: Spike2  

Q. I would like to generate two tones to play as a stimulus and to randomize delivery 
of these waveforms. In addition it would be very useful if I could create new tones 
on-line to replace either of the existing outputs. 

A. The attached script, RandomiseTones.s2s, is an example script that allows the 
user to create a tone A and tone B output waveform and then play these in a 
randomised order during sampling. From the script toolbar the user can generate 
a new tone during sampling and replace the existing tone outputs if required. 

Scripts: Signal  
Q. When stepping through a data file in Signal, it would be very nice if the data was 

automatically optimised for each frame. Is there a setting for this? 

A. The attached script, AutoOptimise.sgs, is an example script that will 
automatically optimise the Y-axis display for each frame off-line when stepping 
through frames in a data file. We plan to add this feature in a future release. 

Did you know…? 
The Micro1401-3 can be expanded up to a maximum of 128 channels using the new 
ADC64 top-box.  

The HumRemove.s2s script, available from our website, can be used in conjunction with 
a 50Hz mains synchronization generator to reduce 50Hz mains noise. Please contact us 
for details. 

'$RandomiseTones.s2s|Example script that allows the user to specify tone waveforms for output and then
'plays them from DAC0 of the 1401 in random order for a specified no. of repeats. You can load in new
'tones to replace the existing ones during sampling (although they must be of the same size/frequency or smaller)
'TextMarks are added to the file to show the parameters of the current tones in use.

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 09/12/09

var data%;                      'Handle of new data file (Spike2 sets it to 0)
var sTime;                      'The last time we looked at the idle routine
var output%;                    'File to hold generated waveforms
var noPointsA%,noPointsB%;    'no. of data points in waveforms
var LengthA;                   'Default output length
var LengthB;
var maxLengthA,maxLengthB;   'Maximum allowed length for waveforms loaded during sampling
var freqA%:=1000;              'Default tone frequency
var freqB%:=1000;
var maxFreqA%,maxFreqB%;    'Max. allowed freq. for waveforms loaded  during sampling
var RiseA:=0.1;                 'Tone A parameters
var PlatA:=0.6;
var FallA:=0.1;          
var RiseB:=0.1;                 'Tone B parameters
var PlatB:=0.6;
var FallB:=0.1;           
var stims$[100];                'Array for stimulus types
var randoms[100];              'Array for random numbers
var c%;                          'counter for no. repeats
var tones$[2];                   'Array for tone names
var reps%;                      'No. of repeats
var toneInfo$;                  'String to hold tone parameters
var type1401%;                 'Check type of 1401 for memory limits
var maxSpace,spaceA,spaceB;  'Space in 1401 memory and size of each output
var path$;
var vers%;

vers%:= App(-1);
if vers% < 600 then
    Message("This script requires Spike2 version 6 or version 7");Halt;endif;

type1401%:= U1401Open();         'Check 1401 type
U1401Close();
if type1401% < 0 then
    Message("No 1401 detected! Check plugged in and powered");
    Halt;endif;

if vers% >= 600 and vers% < 700 then
    path$:= "C:\\Spike6\\data\\outputs.smr";
else
    path$:= "C:\\Spike7\\data\\outputs.smr";
endif;

HideAll();                      'Hide all unwanted bars
ToolbarVisible(1);              'Make toolbar visible always
CreateWave%();
New%();                         'Set up new sampling window
DoToolbar();                    'Do the toolbar
RestoreAll();                   'Restore all bars
Halt;

Func CreateWave%();
var toneA$,toneB$;
var chanA%,chanB%;
var ok%;
var EnvA$, EnvB$;
var ToneAOut%;
var ToneBOut%;
var noReps%;

DlgCreate("Tone settings");  'Start new dialog for tone parameters
DlgText("Tone A parameters",14,1);
DlgInteger(1,"Sine freq.",50,40000,16,2);
DlgReal(2,"Rise time",0.000000,10.000000,16,3);
DlgReal(3,"Plateau",0.001,10.000000,16,4);
DlgReal(4,"Fall time",0.000000,10.000000,16,5);
DlgText("Tone B parameters",30,1);  
DlgInteger(5,"Sine freq.",50,40000,32,2);
DlgReal(6,"Rise time",0.000000,10.000000,32,3);
DlgReal(7,"Plateau",0.001,10.000000,32,4);
DlgReal(8,"Fall time",0.000000,10.000000,32,5);
    
DlgInteger(9,"No. of repeats",1,100,22,8);    
    
DlgButton(0,"Cancel");
DlgButton(1,"OK");
ok% := DlgShow(freqA%,RiseA,PlatA,FallA,freqB%,RiseB,PlatB,FallB,noReps%);   'Settings for tone characteristics

If ok%<1 then
	Message("Script cancelled");
    RestoreAll();
	Halt
endif

LengthA:= RiseA+PlatA+fallA;   
LengthB:= RiseB+PlatB+fallB;     
    
toneInfo$:= Print$("Tone A: %dHz,%0.1f Rise,%0.1f Plat.,%0.1f Fall,%0.1f length|Tone B: %dHz,%0.1f Rise,%0.1f Plat.,%0.1f Fall,%0.1f length",freqA%,riseA,platA,fallA,lengthA,freqB%,riseB,platB,fallB,lengthB);

maxLengthA:= LengthA;        'Get length and freq of original waveforms to check against online updates
maxLengthB:= LengthB;
maxFreqA%:= freqA%;
maxFreqB%:= freqB%;

reps%:= noReps%*2;           'no of repeats for each output
tones$[0]:= "a";
tones$[1]:= "b";
ArrConst(stims$[:noReps%],tones$[0]);       'Fill first n elements with type 1
ArrConst(stims$[noReps%:noReps%],tones$[1]);     '..next n with type 2..

Rand(randoms[:reps%]);	          'Fill randoms with random numbers
ArrSort(randoms[:reps%],0,stims$[:reps%]); 'Sort arrays for random stimulus output

output%:=FileNew(7,0,1,1,LengthA,32);'New file holding output data
toneA$:=Print$("Wsin(%d,0)*5",FreqA%); 'Create a virtual channel and make a waveform
EnvA$:=Print$("WEnv(%f,%f,%f)*",RiseA,PlatA,FallA);
toneA$:=EnvA$+toneA$;
ToneAOut%:=VirtualChan(0,toneA$,0,0.0001,0);  
chanA%:=ChanSave(ToneAOut%,1);   'Save to unused channel
ChanDelete(ToneAOut%);
ChanShow(chanA%);
FileSaveAs("C:\\Spike7\\data\\outputs.smr",-1,1);     'Save file so that we can load waveforms for output
FileClose(0,-1);
FileOpen("C:\\Spike7\\data\\outputs.smr",0);
noPointsA%:= 1/BinSize(chanA%)*LengthA;          'Get no. of data points in waveform
spaceA:=PlayWaveAdd("a","Tone A",0,0.0,LengthA,chanA%,1);
FileClose(0,-1);

output%:=FileNew(7,0,1,1,LengthB,32);'New file holding output data
toneB$:=Print$("Wsin(%d,0)*5",FreqB%); 'Create a virtual channel and make a waveform
EnvB$:=Print$("WEnv(%f,%f,%f)*",RiseB,PlatB,FallB);
toneB$:=EnvB$+toneB$;
ToneBOut%:=VirtualChan(0,toneB$,0,0.0001,0);  
chanB%:=ChanSave(ToneBOut%,1);   'Save to unused channel
ChanDelete(ToneBOut%);
ChanShow(chanB%);
FileSaveAs(path$,-1,1);     'Save data file so that we can load waveforms for output
FileClose(0,-1);
FileOpen(path$,0);
noPointsB%:= 1/BinSize(chanB%)*LengthB;
spaceB:=PlayWaveAdd("b","Tone B",0,0.0,LengthB,chanB%,1);
FileClose(0,-1);
return 1
end;

func New%()                     'New sampling window

if type1401% = 0 or 1 or 2 or 4 then       'If not a Power1401
    maxSpace:= 740000;         'Set limit for waveform size
if (spaceA+spaceB) > maxSpace then      'Check output size against available memory
        Message("Not enough memory in this 1401 for the selected waveform output");
        RestoreAll();
    Halt;endif;
endif;

View(LogHandle());              'Make log view the current view
EditSelectAll();                'Select all text in log view
EditClear();                    'Delete it
Window(0,80,100,100);           'Display it at the bottom of the screen
WindowVisible(1);               'Make it visible
if data%>0 then                 'If there is already a data view open...
  View(data%);                  '...close it
  FileClose();                  'No point setting data% to 0...
endif;
SampleTextMark(200);
data%:=FileNew(0,1);            'Open a new data file for sampling
if data%<0 then Message("Unable to open new data file");Halt() endif;
DrawMode(-1,2);                 'Set event draw mode to lines
Window(0,0,100,80);             'Make data window in top bit of screen
XRange(0,10);

FrontView(LogHandle());         'Bring the Log view to the front
FrontView(data%);               'Bring the data view to the front
ToolbarEnable(3,0);             'Disable "Sample stop" button
ToolbarEnable(2,1);             'Disable "Sample stop" button
ToolbarText("Press SAMPLE START to commence sampling");
return 1;
end;

proc DoToolbar()
ToolbarSet(1,"&Quit",Quit%);    'Set up toolbar buttons
ToolbarSet(2,"&Sample start", Start%);
ToolbarSet(3,"&Sample stop", Stop%);
ToolbarSet(5, "&Play tones", Play%);
ToolbarSet(7, "New Tone", NewTone%);
ToolbarSet(9,"&New file", New%);
ToolbarEnable(3,0);             'Disable "Sample stop" button
Toolbar("Press SAMPLE START to commence sampling", 1023);'Wait here until quit
end;

Func NewTone%()        'Create a new tone to load on-line
var toneN$;
var chanN%;
var freqN%:=1000;
var RiseN:= 0.1;
var PlatN:= 0.6;
var FallN:= 0.1;
var ok%;
var EnvN$;
var LengthN;
var ToneNOut%;
var noPoints%;
var repKey$;
var repKey%;
var clearPoints%;
var lenFlag%;
var freqFlag%;

repeat
DlgCreate("New Tone Settings");  'Tone parameters dialog
DlgInteger(1,"Sine freq.",50,40000);
DlgReal(2,"Rise time",0.000000,10.000000);
DlgReal(3,"Plateau",0.001,10.000000);
DlgReal(4,"Fall time",0.000000,10.000000);
DlgList(5,"Replace tone:",tones$[]);
DlgButton(0,"Cancel");
DlgButton(1,"OK");
ok% := DlgShow(freqN%,RiseN,PlatN,FallN,repKey%);   'Settings for tone characteristics

If ok%>0 then
    LengthN:= RiseN+PlatN+fallN;
    
    if repKey% =0 then
        if LengthN > maxLengthA then
            Message("Output length is larger than the play wave area\nmaximum output length allowed is %0.3f",maxLengthA);    'Check that the new tone doesn't exceed area size
            lenFlag%:= 0;
        else
            lenFlag%:= 1;
        endif;
    else
        if LengthN > maxLengthB then
            Message("Output length is larger than the play wave area\nmaximum output length allowed is %0.3f",maxLengthB);
            lenFlag%:= 0;
        else
            lenFlag%:= 1;
        endif;
	endif;
    
    if repKey% =0 then
        if FreqN% > maxFreqA% then
            Message("No. of data points is larger than the play wave area\nmaximum Sine freq. allowed is %d",maxFreqA%);
            freqFlag%:= 0;
        else
            freqFlag%:= 1;
        endif;
    else
        if FreqN% > maxFreqB% then
            Message("No. of data points is larger than the play wave area\nmaximum Sine freq. allowed is %d",maxFreqB%);
            freqFlag%:= 0;
        else
            freqFlag%:= 1;
        endif;
	endif;
else
    lenFlag%:=1;
        freqFlag%:=1;
    endif;
    Until lenFlag%=1 and freqFlag%=1;
if ok% > 0 then
    if repKey% =0 then                                                                                                         
        repKey$:= "a";
        clearPoints%:= noPointsA%;
        freqA%:= freqN%;
        RiseA:= RiseN;
        PlatA:= PlatN ;
        FallA:= FallN;
        LengthA:= LengthN;
    else
        repKey$:= "b";
        clearPoints%:= noPointsB%;
        freqB%:= freqN%;
        RiseB:= RiseN;
        PlatB:= PlatN ;
        FallB:= FallN;
        LengthB:= LengthN;
    endif;

    toneInfo$:= Print$("Tone A: %dHz,%0.1f Rise,%0.1f Plat.,%0.1f Fall,%0.1f length|Tone B: %dHz,%0.1f Rise,%0.1f Plat.,%0.1f Fall,%0.1f length",freqA%,riseA,platA,fallA,lengthA,freqB%,riseB,platB,fallB,lengthB);

    var blank%[clearPoints%];
    PlayWaveCopy(repKey$,blank%[]);

    output%:=FileNew(7,0,1,1,LengthN,32);'New file holding output data
    Window(25,0,75,50);

    toneN$:=Print$("Wsin(%d,0)*5",FreqN%); 'Create a virtual channel and make a waveform
    EnvN$:=Print$("WEnv(%f,%f,%f)*",RiseN,PlatN,FallN);
    toneN$:=EnvN$+toneN$;
    ToneNOut%:=VirtualChan(0,toneN$,0,0.0001,0); 

    chanN%:=ChanSave(ToneNOut%,1);   'Save to unused channel
    ChanDelete(ToneNOut%);
    ChanShow(chanN%);
    FileSaveAs(path$,-1,1);          'Save file so we can load as a waveform
    FileClose(0,-1);
    FileOpen(path$,0);
    noPoints%:= 1/BinSize(chanN%)*LengthN;
    var data[noPoints%];
    ChanData(1,data[],0.0,LengthN);   'Get waveform data as an array
    FileClose(0,-1);
    PlayWaveCopy(repKey$,data[]);     'Copy to the selected play wave area

    if repKey% = 0 then
        noPointsA%:= noPoints%;
    else
        noPointsB%:= noPoints%;
    endif;
endif;
return 1;
end;

Func Quit%()                    'If "Quit" is pressed
SampleStop();                   'Stop sampling
RestoreAll();
return 0;                       'leave toolbar
end;

Func Start%()                   'If "Start" is pressed
sTime := 0;                     'set start of anaylsis time
SampleStart();                  'Start sampling
ToolbarEnable(4,0);             'Disable "New file" button
ToolbarEnable(3,1);             'Enable "Sample stop" button
ToolbarEnable(2,0);             'Disable "Sample start" button
ToolbarEnable(1,0);             'Disable "Quit" button
ToolbarText("Press SAMPLE STOP to stop sampling");
return 1;                       'Stay with toolbar
end;

Func Stop%()                    'If "Stop" is pressed
SampleStop();                   'Stop sampling
if (data%>0) and                'if no data in file, stop will close it
   (ViewKind(data%) <> 0) then data% := 0 endif;
ToolbarEnable(4,1);             'Enable "New file" button
ToolbarEnable(3,0);             'Disable "Sample stop" button
ToolbarEnable(1,1);             'Enable "Quit" button
ToolbarText("Press FILE NEW to capture more data");
return 1;                       'Stay in toolbar
end;

Func Play%()
c%:= 0;
ToolbarSet(0,"",Idle%);         'Call Idle%() whenever there is free time
ToolbarSet(5,"S&top tones",StopPlay%);      'Set a stop button
SampleText(toneInfo$);         'Set textmark holding tone information
return 1;
end;

Func StopPlay%();
ToolbarClear(0);           'Stop Idle routine
PlayWaveStop(1);          'Stop wave output
ToolbarSet(5,"&Play tones",Play%);      'Set play button
return 1;
end;

Func Idle%()                    'Idle routine runs when system has time
var eTime;
if ViewKind(data%)=0 then       'This checks that the data view exists...
   View(data%);                 '...as this gives an error if it doesn't
    eTime:=Maxtime();
    if (eTime > sTime) and c% < reps% then 'Check for no. of repeats
        SampleKey(stims$[c%]);         'Start next waveform output
        if stims$[c%]= "a" then
            sTime:=eTime+lengthA;   'Wait for current output to finish
        else
            sTime:=eTime+lengthB;
        endif;
        c%+=1;        'Counter for repeats
        if c% = reps% then           
            ToolbarSet(5,"&Play tones",Play%);  'if finished, set Play button to go again.  
        endif;
      endif;
    endif;
   
return 1;                       'Stay in toolbar
end;

'You can find this code in the App() on-line help.
var gFloat%[20];                'global for floating window states
proc HideAll()
var i%;
gFloat%[0] := App(-2);          'number of windows
for i% := 1 to gFloat%[0] do    'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do    'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end




CED
RandomiseTones.s2s


'$AutoOptimise|Script that automatically optimises the Y-axis for each frame as you step through a data file.

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 09/12/09

var file%;             'View handle for file
var curr%;            'Current data frame
var last%;             'Last frame we looked at

file%:=ViewKind();         'Get handle of data file

if file% = 0 or 4 then       'check for a memory or data view 
Optimise(-1);                 'Optimise current frame
    DoToolbar%();
else
    Message("File is not a file or memory view");    'Not a suitable file
    Halt;endif;


Func DoToolbar%();
ToolbarSet(0,"",AutoOpt%);
ToolbarSet(1,"&Quit");
return Toolbar("Auto optimise script",1023);       'Script toolbar
end;

Func AutoOpt%()
    curr%:= View(file%).Frame();      'Get the current frame
    if curr% <> last% then          'If it has changed since we last looked
        Optimise(-1);                   'Optimise
        last%:= curr%                'Update last frame count
    endif;
return 1;
end;

CED
AutoOptimise.sgs

CED
You can change the assumed voltage settings for the 1401 from the Edit preferences dialog. Micro1401-3 and Power1401 mk II units can be switched between ±5 and ±10V range using the Try1401 software



 

Recent questions   
Q. I would like to process every nth frame to an average in Signal to form a grand 

average of every nth response. In this case, I am using a simple square pulse 
stimulus, but I would like to only add every 5th response to my average window.  

A. Using Signal multiple states it is possible to set-up an average that only 
processes frames of a certain state. For this example we can create an additional 
state to represent every 5th output pulse used, and only average the response of 
that particular state. 

 The multiple states checkbox in the General tab of the sampling configuration 
enables the States tab. For our example above we can select Dynamic outputs 
from the State variation list, set Number of extra states to 1, which will be used to 
represent our 5th output, and Ordering to Protocols. Click the Protocols… button 
to set a protocol that will run during sampling. As in the example below, simply 
set Step 1 to State 0, Repeats to 4 and Next to 2. For Step 2 set State 1, 
Repeats 1 and Next to go back to Step 1 in the protocol. Selecting the Run 
protocol automatically at start option will start the protocol running as soon as 
sampling begins.  

 
  

 

 

 

 

 

States tab and Protocol setup dialog 

In the Configure pulses dialog from the Outputs tab, simply copy the existing 
pulse output from Basic 0 to State1 using the Copy button in the dialog.  

When sampling with multiple states you can then select which frame states 
should be included in the average in the Process dialog, shown below with state 
1 selected. In this example, every 5th output is marked as state 1 and is added to 
the average window. 

 
Process settings dialog 

User group      
Try the new CED Forums bulletin board for software and hardware support 

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 
All Trademarks are acknowledged to be the Trademarks of the registered holders. 

